

## Chapter 3

# Expressions

In this chapter, we describe the syntax and informal semantics of Haskell *expressions*, including their translations into the Haskell kernel, where appropriate. Except in the case of `let` expressions, these translations preserve both the static and dynamic semantics. Free variables and constructors used in these translations always refer to entities defined by the `Prelude`. For example, “`concatMap`” used in the translation of list comprehensions (Section 3.11) means the `concatMap` defined by the `Prelude`, regardless of whether or not the identifier “`concatMap`” is in scope where the list comprehension is used, and (if it is in scope) what it is bound to.

In the syntax that follows, there are some families of nonterminals indexed by precedence levels (written as a superscript). Similarly, the nonterminals *op*, *varop*, and *conop* may have a double index: a letter *l*, *r*, or *n* for left-, right- or non-associativity and a precedence level. A precedence-level variable *i* ranges from 0 to 9; an associativity variable *a* varies over {*l*, *r*, *n*}. For example

$$aexp \quad \rightarrow \quad ( exp^{i+1} \ qop^{(a,i)} )$$

actually stands for 30 productions, with 10 substitutions for *i* and 3 for *a*.

$$\begin{array}{ll}
 exp & \rightarrow \quad exp^0 \quad :: \ [context \Rightarrow] \ type & \text{(expression type signature)} \\
 & \quad | \quad exp^0 \\
 exp^i & \rightarrow \quad exp^{i+1} \ [qop^{(n,i)} \ exp^{i+1}] \\
 & \quad | \quad lexp^i \\
 & \quad | \quad rexp^i \\
 lexp^i & \rightarrow \quad (lexp^i \ | \ exp^{i+1}) \ qop^{(l,i)} \ exp^{i+1} \\
 lexp^6 & \rightarrow \quad - \ exp^7
 \end{array}$$

$rexp^i$	$\rightarrow$	$exp^{i+1} \text{ qop}^{(r,i)} (rexp^i \mid exp^{i+1})$	
$exp^{10}$	$\rightarrow$	$\backslash \text{apat}_1 \dots \text{apat}_n \rightarrow exp$	(lambda abstraction, $n \geq 1$ )
		$\text{let decls in exp}$	(let expression)
		$\text{if exp then exp else exp}$	(conditional)
		$\text{case exp of \{ alts \}}$	(case expression)
		$\text{do \{ stmts \}}$	(do expression)
		$fexp$	
$fexp$	$\rightarrow$	$[fexp] aexp$	(function application)
$aexp$	$\rightarrow$	$qvar$	(variable)
		$gcon$	(general constructor)
		$literal$	
		$( exp )$	(parenthesized expression)
		$( exp_1 , \dots , exp_k )$	(tuple, $k \geq 2$ )
		$[ exp_1 , \dots , exp_k ]$	(list, $k \geq 1$ )
		$[ exp_1 [ , exp_2 ] \dots [ exp_2 ] ]$	(arithmetic sequence)
		$[ exp \mid qual_1 , \dots , qual_n ]$	(list comprehension, $n \geq 1$ )
		$( exp^{i+1} \text{ qop}^{(a,i)} )$	(left section)
		$( lexp^i \text{ qop}^{(l,i)} )$	(left section)
		$( \text{qop}^{(a,i)}_{(-)} exp^{i+1} )$	(right section)
		$( \text{qop}^{(r,i)}_{(-)} rexp^i )$	(right section)
		$qcon \{ fbind_1 , \dots , fbind_n \}$	(labeled construction, $n \geq 0$ )
		$aexp_{(qcon)} \{ fbind_1 , \dots , fbind_n \}$	(labeled update, $n \geq 1$ )

Expressions involving infix operators are disambiguated by the operator's fixity (see Section 4.4.2). Consecutive unparenthesized operators with the same precedence must both be either left or right associative to avoid a syntax error. Given an unparenthesized expression " $x \text{ qop}^{(a,i)} y \text{ qop}^{(b,j)} z$ ", parentheses must be added around either " $x \text{ qop}^{(a,i)} y$ " or " $y \text{ qop}^{(b,j)} z$ " when  $i = j$  unless  $a = b = l$  or  $a = b = r$ .

Negation is the only prefix operator in Haskell; it has the same precedence as the infix  $-$  operator defined in the Prelude (see Section 4.4.2, Figure 4.1).

The grammar is ambiguous regarding the extent of lambda abstractions, let expressions, and conditionals. The ambiguity is resolved by the meta-rule that each of these constructs extends as far to the right as possible.

Sample parses are shown below.

This	Parses as
<code>f x + g y</code>	<code>(f x) + (g y)</code>
<code>- f x + y</code>	<code>(- (f x)) + y</code>
<code>let { ... } in x + y</code>	<code>let { ... } in (x + y)</code>
<code>z + let { ... } in x + y</code>	<code>z + (let { ... } in (x + y))</code>
<code>f x y :: Int</code>	<code>(f x y) :: Int</code>
<code>\ x -&gt; a+b :: Int</code>	<code>\ x -&gt; ((a+b) :: Int)</code>

**A note about parsing.** Expressions that involve the interaction of fixities with the let/lambda meta-rule may be hard to parse. For example, the expression

```
let x = True in x == x == True
```

cannot possibly mean

```
let x = True in (x == x == True)
```

because (`==`) is a non-associative operator; so the expression must parse thus:

```
(let x = True in (x == x)) == True
```

However, implementations may well use a post-parsing pass to deal with fixities, so they may well incorrectly deliver the former parse. Programmers are advised to avoid constructs whose parsing involves an interaction of (lack of) associativity with the let/lambda meta-rule.

For the sake of clarity, the rest of this section shows the syntax of expressions without their precedences.

### 3.1 Errors

Errors during expression evaluation, denoted by  $\perp$ , are indistinguishable by a Haskell program from non-termination. Since Haskell is a non-strict language, all Haskell types include  $\perp$ . That is, a value of any type may be bound to a computation that, when demanded, results in an error. When evaluated, errors cause immediate program termination and cannot be caught by the user. The Prelude provides two functions to directly cause such errors:

```
error      :: String -> a
undefined :: a
```

A call to `error` terminates execution of the program and returns an appropriate error indication to the operating system. It should also display the string in some system-dependent manner. When `undefined` is used, the error message is created by the compiler.

Translations of Haskell expressions use `error` and `undefined` to explicitly indicate where execution time errors may occur. The actual program behavior when an error occurs is up to the implementation. The messages passed to the `error` function in these translations are only suggestions; implementations may choose to display more or less information when an error occurs.

### 3.2 Variables, Constructors, Operators, and Literals

<i>aexp</i>	→	<i>qvar</i>	(variable)
		<i>gcon</i>	(general constructor)
		<i>literal</i>	
<i>gcon</i>	→	( )	
		[ ]	
		( , { , } )	
		<i>qcon</i>	
<i>var</i>	→	<i>varid</i>   ( <i>varsym</i> )	(variable)
<i>qvar</i>	→	<i>qvarid</i>   ( <i>qvarsym</i> )	(qualified variable)
<i>con</i>	→	<i>conid</i>   ( <i>consym</i> )	(constructor)
<i>qcon</i>	→	<i>qconid</i>   ( <i>gconsym</i> )	(qualified constructor)
<i>varop</i>	→	<i>varsym</i>   ` <i>varid</i> `	(variable operator)
<i>qvarop</i>	→	<i>qvarsym</i>   ` <i>qvarid</i> `	(qualified variable operator)
<i>conop</i>	→	<i>consym</i>   ` <i>conid</i> `	(constructor operator)
<i>qconop</i>	→	<i>gconsym</i>   ` <i>qconid</i> `	(qualified constructor operator)
<i>op</i>	→	<i>varop</i>   <i>conop</i>	(operator)
<i>qop</i>	→	<i>qvarop</i>   <i>qconop</i>	(qualified operator)
<i>gconsym</i>	→	•   <i>qconsym</i>	

Haskell provides special syntax to support infix notation. An *operator* is a function that can be applied using infix syntax (Section 3.4), or partially applied using a *section* (Section 3.5).

An *operator* is either an *operator symbol*, such as `+` or `$$`, or is an ordinary identifier enclosed in grave accents (backquotes), such as ``op``. For example, instead of writing the prefix application `op x y`, one can write the infix application `x `op` y`. If no fixity declaration is given for ``op`` then it defaults to highest precedence and left associativity (see Section 4.4.2).

Dually, an operator symbol can be converted to an ordinary identifier by enclosing it in parentheses. For example, `(+) x y` is equivalent to `x + y`, and `foldr (*) 1 xs` is equivalent to `foldr (\x y -> x*y) 1 xs`.

Special syntax is used to name some constructors for some of the built-in types, as found in the production for *gcon* and *literal*. These are described in Section 6.1.

An integer literal represents the application of the function `fromInteger` to the appropriate value of type `Integer`. Similarly, a floating point literal stands for an application of `fromRational` to a value of type `Rational` (that is, `Ratio Integer`).

**Translation:** The integer literal  $i$  is equivalent to `fromInteger i`, where `fromInteger` is a method in class `Num` (see Section 6.4.1).

The floating point literal  $f$  is equivalent to `fromRational (n Ratio.% d)`, where `fromRational` is a method in class `Fractional` and `Ratio.%` constructs a rational from two integers, as defined in the `Ratio` library. The integers  $n$  and  $d$  are chosen so that  $n/d = f$ .

### 3.3 Curried Applications and Lambda Abstractions

$fexp$	$\rightarrow$	$[fexp] aexp$	(function application)
$exp$	$\rightarrow$	$\backslash apat_1 \dots apat_n \rightarrow exp$	(lambda abstraction, $n \geq 1$ )

*Function application* is written  $e_1 e_2$ . Application associates to the left, so the parentheses may be omitted in  $(f x) y$ . Because  $e_1$  could be a data constructor, partial applications of data constructors are allowed.

*Lambda abstractions* are written  $\backslash p_1 \dots p_n \rightarrow e$ , where the  $p_i$  are *patterns*. An expression such as  $\backslash x : xs \rightarrow x$  is syntactically incorrect; it may legally be written as  $\backslash (x : xs) \rightarrow x$ .

The set of patterns must be *linear* – no variable may appear more than once in the set.

**Translation:** The following identity holds:

$$\backslash p_1 \dots p_n \rightarrow e = \backslash x_1 \dots x_n \rightarrow \text{case } (x_1, \dots, x_n) \text{ of } (p_1, \dots, p_n) \rightarrow e$$

where the  $x_i$  are new identifiers.

Given this translation combined with the semantics of case expressions and pattern matching described in Section 3.17.3, if the pattern fails to match, then the result is  $\perp$ .

### 3.4 Operator Applications

$exp$	$\rightarrow$	$exp_1 qop exp_2$	
	$ $	$- exp$	(prefix negation)
$qop$	$\rightarrow$	$qvarop   qconop$	(qualified operator)

The form  $e_1 \text{ qop } e_2$  is the infix application of binary operator  $\text{qop}$  to expressions  $e_1$  and  $e_2$ .

The special form  $-e$  denotes prefix negation, the only prefix operator in Haskell, and is syntax for `negate (e)`. The binary  $-$  operator does not necessarily refer to the definition of  $-$  in the Prelude; it may be rebound by the module system. However, unary  $-$  will always refer to the `negate` function defined in the Prelude. There is no link between the local meaning of the  $-$  operator and unary negation.

Prefix negation has the same precedence as the infix operator  $-$  defined in the Prelude (see Table 4.1, p. 57). Because  $e_1 - e_2$  parses as an infix application of the binary operator  $-$ , one must write  $e_1 (-e_2)$  for the alternative parsing. Similarly,  $(-)$  is syntax for  $(\lambda x y \rightarrow x - y)$ , as with any infix operator, and does not denote  $(\lambda x \rightarrow -x)$  – one must use `negate` for that.

**Translation:** The following identities hold:

$$\begin{aligned} e_1 \text{ op } e_2 &= (\text{op}) e_1 e_2 \\ -e &= \text{negate } (e) \end{aligned}$$

### 3.5 Sections

$$\begin{aligned} a \text{ exp} &\rightarrow ( \text{exp}^{i+1} \text{ qop}^{(a,i)} ) && \text{(left section)} \\ &| ( \text{lexp}^i \text{ qop}^{(l,i)} ) && \text{(left section)} \\ &| ( \text{qop}_{(-)}^{(a,i)} \text{ exp}^{i+1} ) && \text{(right section)} \\ &| ( \text{qop}_{(-)}^{(r,i)} \text{ rexp}^i ) && \text{(right section)} \end{aligned}$$

*Sections* are written as  $( \text{op } e )$  or  $( e \text{ op} )$ , where  $\text{op}$  is a binary operator and  $e$  is an expression. Sections are a convenient syntax for partial application of binary operators.

Syntactic precedence rules apply to sections as follows.  $( \text{op } e )$  is legal if and only if  $( x \text{ op } e )$  parses in the same way as  $( x \text{ op } ( e ) )$ ; and similarly for  $( e \text{ op} )$ . For example,  $( * a + b )$  is syntactically invalid, but  $( + a * b )$  and  $( * ( a + b ) )$  are valid. Because  $( + )$  is left associative,  $( a + b + )$  is syntactically correct, but  $( + a + b )$  is not; the latter may legally be written as  $( + ( a + b ) )$ . As another example, the expression

```
(let n = 10 in n +)
```

is invalid because, by the `let/lambda` meta-rule (Section 3), the expression

```
(let n = 10 in n + x)
```

parses as

```
(let n = 10 in (n + x))
```

rather than

```
((let n = 10 in n) + x)
```

Because `-` is treated specially in the grammar, `(- exp)` is not a section, but an application of prefix negation, as described in the preceding section. However, there is a `subtract` function defined in the Prelude such that `(subtract exp)` is equivalent to the disallowed section. The expression `(+ (- exp))` can serve the same purpose.

**Translation:** The following identities hold:

$$\begin{aligned} (op\ e) &= \lambda x \rightarrow x\ op\ e \\ (e\ op) &= \lambda x \rightarrow e\ op\ x \end{aligned}$$

where `op` is a binary operator, `e` is an expression, and `x` is a variable that does not occur free in `e`.

### 3.6 Conditionals

```
exp → if exp1 then exp2 else exp3
```

A *conditional expression* has the form `if e1 then e2 else e3` and returns the value of `e2` if the value of `e1` is `True`, `e3` if `e1` is `False`, and `⊥` otherwise.

**Translation:** The following identity holds:

$$\text{if } e_1 \text{ then } e_2 \text{ else } e_3 = \text{case } e_1 \text{ of } \{ \text{True} \rightarrow e_2 ; \text{False} \rightarrow e_3 \}$$

where `True` and `False` are the two nullary constructors from the type `Bool`, as defined in the Prelude. The type of `e1` must be `Bool`; `e2` and `e3` must have the same type, which is also the type of the entire conditional expression.

### 3.7 Lists

```
exp → exp1 qop exp2
aexp → [ exp1 , ... , expk ] (k ≥ 1)
      | gcon
gcon → [ ]
      | qcon
```

$qcon \rightarrow ( gconsym )$   
 $qop \rightarrow qconop$   
 $qconop \rightarrow gconsym$   
 $gconsym \rightarrow :$

*Lists* are written  $[ e_1 , \dots , e_k ]$ , where  $k \geq 1$ . The list constructor is  $:$ , and the empty list is denoted  $[]$ . Standard operations on lists are given in the Prelude (see Section 6.1.3, and Chapter 8 notably Section 8.2).

**Translation:** The following identity holds:

$$[ e_1 , \dots , e_k ] = e_1 : ( e_2 : ( \dots ( e_k : [] ) ) )$$

where  $:$  and  $[]$  are constructors for lists, as defined in the Prelude (see Section 6.1.3). The types of  $e_1$  through  $e_k$  must all be the same (call it  $t$ ), and the type of the overall expression is  $[ t ]$  (see Section 4.1.2).

The constructor “ $:$ ” is reserved solely for list construction; like  $[]$ , it is considered part of the language syntax, and cannot be hidden or redefined. It is a right-associative operator, with precedence level 5 (Section 4.4.2).

### 3.8 Tuples

$aexp \rightarrow ( exp_1 , \dots , exp_k ) \quad (k \geq 2)$   
 $\quad \quad \quad | \quad qcon$   
 $qcon \rightarrow ( , \{ , \} )$

*Tuples* are written  $( e_1 , \dots , e_k )$ , and may be of arbitrary length  $k \geq 2$ . The constructor for an  $n$ -tuple is denoted by  $( , \dots , )$ , where there are  $n - 1$  commas. Thus  $( a , b , c )$  and  $( , , ) a b c$  denote the same value. Standard operations on tuples are given in the Prelude (see Section 6.1.4 and Chapter 8).

**Translation:**  $( e_1 , \dots , e_k )$  for  $k \geq 2$  is an instance of a  $k$ -tuple as defined in the Prelude, and requires no translation. If  $t_1$  through  $t_k$  are the types of  $e_1$  through  $e_k$ , respectively, then the type of the resulting tuple is  $( t_1 , \dots , t_k )$  (see Section 4.1.2).

### 3.9 Unit Expressions and Parenthesized Expressions

$$\begin{array}{lcl} aexp & \rightarrow & gcon \\ & | & ( exp ) \\ gcon & \rightarrow & ( ) \end{array}$$

The form  $( e )$  is simply a *parenthesized expression*, and is equivalent to  $e$ . The *unit expression*  $( )$  has type  $( )$  (see Section 4.1.2). It is the only member of that type apart from  $\perp$ , and can be thought of as the “nullary tuple” (see Section 6.1.5).

**Translation:**  $( e )$  is equivalent to  $e$ .

### 3.10 Arithmetic Sequences

$$aexp \quad \rightarrow \quad [ exp_1 [ , exp_2 ] \dots [ exp_3 ] ]$$

The *arithmetic sequence*  $[ e_1 , e_2 \dots e_3 ]$  denotes a list of values of type  $t$ , where each of the  $e_i$  has type  $t$ , and  $t$  is an instance of class Enum.

**Translation:** Arithmetic sequences satisfy these identities:

$$\begin{array}{lcl} [ e_1 \dots ] & = & \text{enumFrom } e_1 \\ [ e_1 , e_2 \dots ] & = & \text{enumFromThen } e_1 e_2 \\ [ e_1 \dots e_3 ] & = & \text{enumFromTo } e_1 e_3 \\ [ e_1 , e_2 \dots e_3 ] & = & \text{enumFromThenTo } e_1 e_2 e_3 \end{array}$$

where `enumFrom`, `enumFromThen`, `enumFromTo`, and `enumFromThenTo` are class methods in the class `Enum` as defined in the Prelude (see Figure 6.1, p. 85).

The semantics of arithmetic sequences therefore depends entirely on the instance declaration for the type  $t$ . See Section 6.3.4 for more details of which Prelude types are in Enum and their semantics.

### 3.11 List Comprehensions

$$\begin{array}{lcl} aexp & \rightarrow & [ exp \mid qual_1 , \dots , qual_n ] & \text{(list comprehension, } n \geq 1) \\ qual & \rightarrow & pat <- exp & \text{(generator)} \\ & | & \text{let decls} & \text{(local declaration)} \\ & | & exp & \text{(guard)} \end{array}$$

A *list comprehension* has the form  $[ e \mid q_1, \dots, q_n ], n \geq 1$ , where the  $q_i$  qualifiers are either

- *generators* of the form  $p \leftarrow e$ , where  $p$  is a pattern (see Section 3.17) of type  $t$  and  $e$  is an expression of type  $[ t ]$
- *guards*, which are arbitrary expressions of type `Bool`
- *local bindings* that provide new definitions for use in the generated expression  $e$  or subsequent guards and generators.

Such a list comprehension returns the list of elements produced by evaluating  $e$  in the successive environments created by the nested, depth-first evaluation of the generators in the qualifier list. Binding of variables occurs according to the normal pattern matching rules (see Section 3.17), and if a match fails then that element of the list is simply skipped over. Thus:

```
[ x | xs <- [ [(1,2),(3,4)], [(5,4),(3,2)] ],
      (3,x) <- xs ]
```

yields the list  $[ 4, 2 ]$ . If a qualifier is a guard, it must evaluate to `True` for the previous pattern match to succeed. As usual, bindings in list comprehensions can shadow those in outer scopes; for example:

```
[ x | x <- x, x <- x ] = [ z | y <- x, z <- y ]
```

**Translation:** List comprehensions satisfy these identities, which may be used as a translation into the kernel:

```
[ e | True ]           = [ e ]
[ e | q ]              = [ e | q, True ]
[ e | b, Q ]           = if b then [ e | Q ] else []
[ e | p <- l, Q ]      = let ok p = [ e | Q ]
                        ok _ = []
                        in concatMap ok l
[ e | let decls, Q ] = let decls in [ e | Q ]
```

where  $e$  ranges over expressions,  $p$  over patterns,  $l$  over list-valued expressions,  $b$  over boolean expressions,  $decls$  over declaration lists,  $q$  over qualifiers, and  $Q$  over sequences of qualifiers. `ok` is a fresh variable. The function `concatMap`, and boolean value `True`, are defined in the Prelude.

As indicated by the translation of list comprehensions, variables bound by `let` have fully polymorphic types while those defined by `<-` are lambda bound and are thus monomorphic (see Section 4.5.4).

### 3.12 Let Expressions

$exp \rightarrow \text{let } decls \text{ in } exp$

Let expressions have the general form  $\text{let } \{ d_1 ; \dots ; d_n \} \text{ in } e$ , and introduce a nested, lexically-scoped, mutually-recursive list of declarations (`let` is often called `letrec` in other languages). The scope of the declarations is the expression  $e$  and the right hand side of the declarations. Declarations are described in Chapter 4. Pattern bindings are matched lazily; an implicit  $\sim$  makes these patterns irrefutable. For example,

`let (x,y) = undefined in e`

does not cause an execution-time error until  $x$  or  $y$  is evaluated.

**Translation:** The dynamic semantics of the expression  $\text{let } \{ d_1 ; \dots ; d_n \} \text{ in } e_0$  are captured by this translation: After removing all type signatures, each declaration  $d_i$  is translated into an equation of the form  $p_i = e_i$ , where  $p_i$  and  $e_i$  are patterns and expressions respectively, using the translation in Section 4.4.3. Once done, these identities hold, which may be used as a translation into the kernel:

$$\begin{aligned} \text{let } \{ p_1=e_1 ; \dots ; p_n=e_n \} \text{ in } e_0 &= \text{let } (\sim p_1, \dots, \sim p_n) = (e_1, \dots, e_n) \text{ in } e_0 \\ \text{let } p = e_1 \text{ in } e_0 &= \text{case } e_1 \text{ of } \sim p \rightarrow e_0 \\ &\quad \text{where no variable in } p \text{ appears free in } e_1 \\ \text{let } p = e_1 \text{ in } e_0 &= \text{let } p = \text{fix } (\backslash \sim p \rightarrow e_1) \text{ in } e_0 \end{aligned}$$

where `fix` is the least fixpoint operator. Note the use of the irrefutable patterns  $\sim p$ . This translation does not preserve the static semantics because the use of `case` precludes a fully polymorphic typing of the bound variables. The static semantics of the bindings in a `let` expression are described in Section 4.4.3.

### 3.13 Case Expressions

$$\begin{aligned} exp &\rightarrow \text{case } exp \text{ of } \{ alts \} \\ alts &\rightarrow alt_1 ; \dots ; alt_n && (n \geq 1) \\ alt &\rightarrow pat \rightarrow exp \text{ [where } decls \text{]} \\ &| pat \text{ } gdpat \text{ [where } decls \text{]} \\ &| && (\text{empty alternative}) \\ gdpat &\rightarrow gd \rightarrow exp \text{ [ } gdpat \text{ ]} \\ gd &\rightarrow | exp^0 \end{aligned}$$

A case expression has the general form

$$\text{case } e \text{ of } \{ p_1 \text{ match}_1 ; \dots ; p_n \text{ match}_n \}$$

where each  $match_i$  is of the general form

$$\begin{array}{l} | g_{i1} \quad \rightarrow e_{i1} \\ \dots \\ | g_{im_i} \quad \rightarrow e_{im_i} \\ \text{where } decls_i \end{array}$$

(Notice that in the syntax rule for  $gd$ , the “|” is a terminal symbol, not the syntactic metasymbol for alternation.) Each alternative  $p_i match_i$  consists of a pattern  $p_i$  and its matches,  $match_i$ . Each match in turn consists of a sequence of pairs of guards  $g_{ij}$  and bodies  $e_{ij}$  (expressions), followed by optional bindings ( $decls_i$ ) that scope over all of the guards and expressions of the alternative. An alternative of the form

$$pat \rightarrow exp \text{ where } decls$$

is treated as shorthand for:

$$\begin{array}{l} pat \mid \text{True} \rightarrow exp \\ \text{where } decls \end{array}$$

A case expression must have at least one alternative and each alternative must have at least one body. Each body must have the same type, and the type of the whole expression is that type.

A case expression is evaluated by pattern matching the expression  $e$  against the individual alternatives. The alternatives are tried sequentially, from top to bottom. If  $e$  matches the pattern in the alternative, the guards for that alternative are tried sequentially from top to bottom, in the environment of the case expression extended first by the bindings created during the matching of the pattern, and then by the  $decls_i$  in the `where` clause associated with that alternative. If one of the guards evaluates to `True`, the corresponding right-hand side is evaluated in the same environment as the guard. If all the guards evaluate to `False`, matching continues with the next alternative. If no match succeeds, the result is  $\perp$ . Pattern matching is described in Section 3.17, with the formal semantics of case expressions in Section 3.17.3.

**A note about parsing.** The expression

```
case x of { (a,_) | let b = not a in b :: Bool -> a }
```

is tricky to parse correctly. It has a single unambiguous parse, namely

```
case x of { (a,_) | (let b = not a in b :: Bool) -> a }
```

However, the phrase `Bool -> a` is syntactically valid as a type, and parsers with limited lookahead may incorrectly commit to this choice, and hence reject the program. Programmers are advised, therefore, to avoid guards that end with a type signature – indeed that is why a  $gd$  contains an  $exp^0$  not an  $exp$ .

### 3.14 Do Expressions

<i>exp</i>	→	<code>do { stmts }</code>	(do expression)
<i>stmts</i>	→	<code>stmt<sub>1</sub> ... stmt<sub>n</sub> exp [;]</code>	( $n \geq 0$ )
<i>stmt</i>	→	<code>exp ;</code>	
		<code>pat &lt;- exp ;</code>	
		<code>let decls ;</code>	
		<code>;</code>	(empty statement)

A *do expression* provides a more conventional syntax for monadic programming. It allows an expression such as

```
putStr "x: "    >>
getLine        >>= \l ->
return (words l)
```

to be written in a more traditional way as:

```
do putStr "x: "
  l <- getLine
  return (words l)
```

**Translation:** Do expressions satisfy these identities, which may be used as a translation into the kernel, after eliminating empty *stmts*:

```
do {e}                = e
do {e; stmts}         = e >> do {stmts}
do {p <- e; stmts}    = let ok p = do {stmts}
                        ok _ = fail "... "
                        in e >>= ok
do {let decls; stmts} = let decls in do {stmts}
```

The ellipsis "... " stands for a compiler-generated error message, passed to `fail`, preferably giving some indication of the location of the pattern-match failure; the functions `>>`, `>>=`, and `fail` are operations in the class `Monad`, as defined in the Prelude; and `ok` is a fresh identifier.

As indicated by the translation of `do`, variables bound by `let` have fully polymorphic types while those defined by `<-` are lambda bound and are thus monomorphic.

### 3.15 Datatypes with Field Labels

A datatype declaration may optionally define field labels (see Section 4.2.1). These field labels can be used to construct, select from, and update fields in a manner that is independent of the overall structure of the datatype.

Different datatypes cannot share common field labels in the same scope. A field label can be used at most once in a constructor. Within a datatype, however, a field label can be used in more than one constructor provided the field has the same typing in all constructors. To illustrate the last point, consider:

```
data S = S1 { x :: Int } | S2 { x :: Int } -- OK
data T = T1 { y :: Int } | T2 { y :: Bool } -- BAD
```

Here *S* is legal but *T* is not, because *y* is given inconsistent typings in the latter.

### 3.15.1 Field Selection

*aexp* → *qvar*

Field labels are used as selector functions. When used as a variable, a field label serves as a function that extracts the field from an object. Selectors are top level bindings and so they may be shadowed by local variables but cannot conflict with other top level bindings of the same name. This shadowing only affects selector functions; in record construction (Section 3.15.2) and update (Section 3.15.3), field labels cannot be confused with ordinary variables.

**Translation:** A field label *f* introduces a selector function defined as:

$$f \ x = \text{case } x \text{ of } \{ C_1 \ p_{11} \ \dots \ p_{1k} \ \rightarrow \ e_1 ; \dots ; C_n \ p_{n1} \ \dots \ p_{nk} \ \rightarrow \ e_n \}$$

where  $C_1 \dots C_n$  are all the constructors of the datatype containing a field labeled with *f*,  $p_{ij}$  is *y* when *f* labels the *j*th component of  $C_i$  or `_` otherwise, and  $e_i$  is *y* when some field in  $C_i$  has a label of *f* or `undefined` otherwise.

### 3.15.2 Construction Using Field Labels

*aexp* → *qcon* { *fbind*<sub>1</sub> , ... , *fbind*<sub>*n*</sub> } (labeled construction,  $n \geq 0$ )  
*fbind* → *qvar* = *exp*

A constructor with labeled fields may be used to construct a value in which the components are specified by name rather than by position. Unlike the braces used in declaration lists, these are not subject to layout; the { and } characters must be explicit. (This is also true of field updates and field patterns.) Construction using field labels is subject to the following constraints:

- Only field labels declared with the specified constructor may be mentioned.
- A field label may not be mentioned more than once.
- Fields not mentioned are initialized to  $\perp$ .

- A compile-time error occurs when any strict fields (fields whose declared types are prefixed by ! ) are omitted during construction. Strict fields are discussed in Section 4.2.1.

The expression  $F \{ \}$ , where  $F$  is a data constructor, is legal *whether or not*  $F$  was declared with *record syntax* (provided  $F$  has no strict fields – see the third bullet above); it denotes  $F \perp_1 \dots \perp_n$ , where  $n$  is the arity of  $F$ .

**Translation:** In the binding  $f = v$ , the field  $f$  labels  $v$ .

$$C \{ bs \} = C (pick_1^C bs \text{ undefined}) \dots (pick_k^C bs \text{ undefined})$$

where  $k$  is the arity of  $C$ .

The auxiliary function  $pick_i^C bs d$  is defined as follows:

If the  $i$ th component of a constructor  $C$  has the field label  $f$ , and if  $f = v$  appears in the binding list  $bs$ , then  $pick_i^C bs d$  is  $v$ . Otherwise,  $pick_i^C bs d$  is the default value  $d$ .

### 3.15.3 Updates Using Field Labels

$$aexp \rightarrow aexp_{\langle qcon \rangle} \{ fbind_1, \dots, fbind_n \} \quad (\text{labeled update, } n \geq 1)$$

Values belonging to a datatype with field labels may be non-destructively updated. This creates a new value in which the specified field values replace those in the existing value. Updates are restricted in the following ways:

- All labels must be taken from the same datatype.
- At least one constructor must define all of the labels mentioned in the update.
- No label may be mentioned more than once.
- An execution error occurs when the value being updated does not contain all of the specified labels.

**Translation:** Using the prior definition of  $pick$ ,

$$e \{ bs \} = \text{case } e \text{ of}$$

$$C_1 v_1 \dots v_{k_1} \rightarrow C_1 (pick_1^{C_1} bs v_1) \dots (pick_{k_1}^{C_1} bs v_{k_1})$$

$$\dots$$

$$C_j v_1 \dots v_{k_j} \rightarrow C_j (pick_1^{C_j} bs v_1) \dots (pick_{k_j}^{C_j} bs v_{k_j})$$

$$\_ \rightarrow \text{error "Update error"}$$

where  $\{C_1, \dots, C_j\}$  is the set of constructors containing all labels in  $bs$ , and  $k_i$  is the arity of  $C_i$ .

Here are some examples using labeled fields:

```
data T      = C1 {f1,f2 :: Int}
           | C2 {f1 :: Int,
                f3,f4 :: Char}
```

Expression	Translation
C1 {f1 = 3}	C1 3 undefined
C2 {f1 = 1, f4 = 'A', f3 = 'B'}	C2 1 'B' 'A'
x {f1 = 1}	case x of C1 _ f2    -> C1 1 f2 C2 _ f3 f4 -> C2 1 f3 f4

The field `f1` is common to both constructors in `T`. This example translates expressions using constructors in field-label notation into equivalent expressions using the same constructors without field labels. A compile-time error will result if no single constructor defines the set of field labels used in an update, such as `x {f2 = 1, f3 = 'x'}`.

### 3.16 Expression Type-Signatures

$$exp \quad \rightarrow \quad exp :: [context \Rightarrow] type$$

*Expression type-signatures* have the form  $e :: t$ , where  $e$  is an expression and  $t$  is a type (Section 4.1.2); they are used to type an expression explicitly and may be used to resolve ambiguous typings due to overloading (see Section 4.3.4). The value of the expression is just that of  $exp$ . As with normal type signatures (see Section 4.4.1), the declared type may be more specific than the principal type derivable from  $exp$ , but it is an error to give a type that is more general than, or not comparable to, the principal type.

#### Translation:

$$e :: t = \text{let } \{v :: t; v = e\} \text{ in } v$$

### 3.17 Pattern Matching

*Patterns* appear in lambda abstractions, function definitions, pattern bindings, list comprehensions, do expressions, and case expressions. However, the first five of these ultimately translate into case expressions, so defining the semantics of pattern matching for case expressions is sufficient.

#### 3.17.1 Patterns

Patterns have this syntax:

$pat$	$\rightarrow$	$var + integer$	(successor pattern)
		$pat^0$	
$pat^i$	$\rightarrow$	$pat^{i+1} [qconop^{(n,i)} pat^{i+1}]$	
		$lpat^i$	
		$rpat^i$	
$lpat^i$	$\rightarrow$	$(lpat^i \mid pat^{i+1}) qconop^{(l,i)} pat^{i+1}$	
$lpat^6$	$\rightarrow$	$-(integer \mid float)$	(negative literal)
$rpat^i$	$\rightarrow$	$pat^{i+1} qconop^{(r,i)} (rpat^i \mid pat^{i+1})$	
$pat^{10}$	$\rightarrow$	$apat$	
		$gcon\ apat_1 \dots apat_k$	(arity $gcon = k, k \geq 1$ )
$apat$	$\rightarrow$	$var\ [@\ apat]$	(as pattern)
		$gcon$	(arity $gcon = 0$ )
		$qcon\ \{ fpat_1, \dots, fpat_k \}$	(labeled pattern, $k \geq 0$ )
		$literal$	
		$\_$	(wildcard)
		$(\ pat )$	(parenthesized pattern)
		$(\ pat_1, \dots, pat_k )$	(tuple pattern, $k \geq 2$ )
		$[\ pat_1, \dots, pat_k ]$	(list pattern, $k \geq 1$ )
		$\sim\ apat$	(irrefutable pattern)
$fpat$	$\rightarrow$	$qvar = pat$	

The arity of a constructor must match the number of sub-patterns associated with it; one cannot match against a partially-applied constructor.

All patterns must be *linear* – no variable may appear more than once. For example, this definition is illegal:

```
f (x,x) = x  -- ILLEGAL; x used twice in pattern
```

Patterns of the form  $var@pat$  are called *as-patterns*, and allow one to use  $var$  as a name for the value being matched by  $pat$ . For example,

```
case e of { xs@(x:rest) -> if x==0 then rest else xs }
```

is equivalent to:

```
let { xs = e } in
  case xs of { (x:rest) -> if x==0 then rest else xs }
```

Patterns of the form  $\_$  are *wildcards* and are useful when some part of a pattern is not referenced on the right-hand side. It is as if an identifier not used elsewhere were put in its place. For example,

```
case e of { [x,_,_] -> if x==0 then True else False }
```

is equivalent to:

```
case e of { [x,y,z] -> if x==0 then True else False }
```

### 3.17.2 Informal Semantics of Pattern Matching

Patterns are matched against values. Attempting to match a pattern can have one of three results: it may *fail*; it may *succeed*, returning a binding for each variable in the pattern; or it may *diverge* (i.e. return  $\perp$ ). Pattern matching proceeds from left to right, and outside to inside, according to the following rules:

1. Matching the pattern *var* against a value *v* always succeeds and binds *var* to *v*.
2. Matching the pattern  $\sim$  *apat* against a value *v* always succeeds. The free variables in *apat* are bound to the appropriate values if matching *apat* against *v* would otherwise succeed, and to  $\perp$  if matching *apat* against *v* fails or diverges. (Binding does *not* imply evaluation.)

Operationally, this means that no matching is done on a  $\sim$  *apat* pattern until one of the variables in *apat* is used. At that point the entire pattern is matched against the value, and if the match fails or diverges, so does the overall computation.

3. Matching the wildcard pattern  $\_$  against any value always succeeds, and no binding is done.
4. Matching the pattern *con pat* against a value, where *con* is a constructor defined by `newtype`, depends on the value:
  - If the value is of the form *con v*, then *pat* is matched against *v*.
  - If the value is  $\perp$ , then *pat* is matched against  $\perp$ .

That is, constructors associated with `newtype` serve only to change the type of a value.

5. Matching the pattern *con pat*<sub>1</sub> ... *pat*<sub>*n*</sub> against a value, where *con* is a constructor defined by `data`, depends on the value:
  - If the value is of the form *con v*<sub>1</sub> ... *v*<sub>*n*</sub>, sub-patterns are matched left-to-right against the components of the data value; if all matches succeed, the overall match succeeds; the first to fail or diverge causes the overall match to fail or diverge, respectively.
  - If the value is of the form *con'* *v*<sub>1</sub> ... *v*<sub>*m*</sub>, where *con* is a different constructor to *con'*, the match fails.
  - If the value is  $\perp$ , the match diverges.
6. Matching against a constructor using labeled fields is the same as matching ordinary constructor patterns except that the fields are matched in the order they are named in the field list. All fields listed must be declared by the constructor; fields may not be named more than once. Fields not named by the pattern are ignored (matched against  $\_$ ).

7. Matching a numeric, character, or string literal pattern  $k$  against a value  $v$  succeeds if  $v == k$ , where  $==$  is overloaded based on the type of the pattern. The match diverges if this test diverges.

The interpretation of numeric literals is exactly as described in Section 3.2; that is, the overloaded function `fromInteger` or `fromRational` is applied to an `Integer` or `Rational` literal (resp) to convert it to the appropriate type.

8. Matching an  $n+k$  pattern (where  $n$  is a variable and  $k$  is a positive integer literal) against a value  $v$  succeeds if  $x \geq k$ , resulting in the binding of  $n$  to  $x - k$ , and fails otherwise. Again, the functions  $\geq$  and  $-$  are overloaded, depending on the type of the pattern. The match diverges if the comparison diverges.

The interpretation of the literal  $k$  is the same as in numeric literal patterns, except that only integer literals are allowed.

9. Matching an as-pattern  $var @ \text{apat}$  against a value  $v$  is the result of matching  $\text{apat}$  against  $v$ , augmented with the binding of  $var$  to  $v$ . If the match of  $\text{apat}$  against  $v$  fails or diverges, then so does the overall match.

Aside from the obvious static type constraints (for example, it is a static error to match a character against a boolean), the following static class constraints hold:

- An integer literal pattern can only be matched against a value in the class `Num`.
- A floating literal pattern can only be matched against a value in the class `Fractional`.
- An  $n+k$  pattern can only be matched against a value in the class `Integral`.

Many people feel that  $n+k$  patterns should not be used. These patterns may be removed or changed in future versions of Haskell.

It is sometimes helpful to distinguish two kinds of patterns. Matching an *irrefutable pattern* is non-strict: the pattern matches even if the value to be matched is  $\perp$ . Matching a *refutable pattern* is strict: if the value to be matched is  $\perp$  the match diverges. The irrefutable patterns are as follows: a variable, a wildcard,  $N \text{ apat}$  where  $N$  is a constructor defined by `newtype` and  $\text{apat}$  is irrefutable (see Section 4.2.3),  $var @ \text{apat}$  where  $\text{apat}$  is irrefutable, or of the form  $\sim \text{apat}$  (whether or not  $\text{apat}$  is irrefutable). All other patterns are *refutable*.

Here are some examples:

1. If the pattern  $[ 'a', 'b' ]$  is matched against  $[ 'x', \perp ]$ , then  $'a'$  *fails* to match against  $'x'$ , and the result is a failed match. But if  $[ 'a', 'b' ]$  is matched against  $[ \perp, 'x' ]$ , then attempting to match  $'a'$  against  $\perp$  causes the match to *diverge*.
2. These examples demonstrate refutable vs. irrefutable matching:

$$\begin{aligned} (\backslash \sim (x, y) \rightarrow 0) \perp &\Rightarrow 0 \\ (\backslash (x, y) \rightarrow 0) \perp &\Rightarrow \perp \end{aligned}$$

$$\begin{aligned}
(\backslash \sim[x] \rightarrow 0) [] &\Rightarrow 0 \\
(\backslash \sim[x] \rightarrow x) [] &\Rightarrow \perp \\
\\
(\backslash \sim[x, \sim(a,b)] \rightarrow x) [(0,1), \perp] &\Rightarrow (0,1) \\
(\backslash \sim[x, (a,b)] \rightarrow x) [(0,1), \perp] &\Rightarrow \perp \\
\\
(\backslash (x:xs) \rightarrow x:x:xs) \perp &\Rightarrow \perp \\
(\backslash \sim(x:xs) \rightarrow x:x:xs) \perp &\Rightarrow \perp:\perp:\perp
\end{aligned}$$

3. Consider the following declarations:

```

newtype N = N Bool
data D = D !Bool

```

These examples illustrate the difference in pattern matching between types defined by `data` and `newtype`:

$$\begin{aligned}
(\backslash (N \text{ True}) \rightarrow \text{True}) \perp &\Rightarrow \perp \\
(\backslash (D \text{ True}) \rightarrow \text{True}) \perp &\Rightarrow \perp \\
(\backslash \sim(D \text{ True}) \rightarrow \text{True}) \perp &\Rightarrow \text{True}
\end{aligned}$$

Additional examples may be found in Section 4.2.3.

Top level patterns in case expressions and the set of top level patterns in function or pattern bindings may have zero or more associated *guards*. A guard is a boolean expression that is evaluated only after all of the arguments have been successfully matched, and it must be true for the overall pattern match to succeed. The environment of the guard is the same as the right-hand-side of the case-expression alternative, function definition, or pattern binding to which it is attached.

The guard semantics have an obvious influence on the strictness characteristics of a function or case expression. In particular, an otherwise irrefutable pattern may be evaluated because of a guard. For example, in

```

f :: (Int, Int, Int) -> [Int] -> Int
f ~ (x, y, z) [a] | (a == y) = 1

```

both `a` and `y` will be evaluated by `==` in the guard.

### 3.17.3 Formal Semantics of Pattern Matching

The semantics of all pattern matching constructs other than `case` expressions are defined by giving identities that relate those constructs to `case` expressions. The semantics of `case` expressions themselves are in turn given as a series of identities, in Figures 3.1 and 3.2. Any implementation should behave so that these identities hold; it is not expected that it will use them directly, since that would generate rather inefficient code.

```

(a) case e of { alts } = (\v -> case v of { alts }) e
    where v is a new variable
(b) case v of { p1 match1; ... ; pn matchn }
    = case v of { p1 match1 ;
          _ -> ... case v of {
              pn matchn ;
          _ -> error "No match" }...}

    where each matchi has the form:
      | gi,1 -> ei,1 ; ... ; | gi,mi -> ei,mi where { declsi }
(c) case v of { p | g1 -> e1 ; ...
          | gn -> en where { decls }
          _ -> e' }
    = case e' of
      { y -> (where y is a new variable)
        case v of {
          p -> let { decls } in
            if g1 then e1 ... else if gn then en else y ;
          _ -> y }}
(d) case v of { ~p -> e; _ -> e' }
    = (\x1 ... xn -> e) (case v of { p -> x1 }) ... (case v of { p -> xn })
    where x1, ..., xn are all the variables in p
(e) case v of { x@p -> e; _ -> e' }
    = case v of { p -> ( \ x -> e ) v ; _ -> e' }
(f) case v of { _ -> e; _ -> e' } = e

```

Figure 3.1: Semantics of Case Expressions, Part 1

In Figures 3.1 and 3.2: *e*, *e'* and *e<sub>i</sub>* are expressions; *g* and *g<sub>i</sub>* are boolean-valued expressions; *p* and *p<sub>i</sub>* are patterns; *v*, *x*, and *x<sub>i</sub>* are variables; *K* and *K'* are algebraic datatype (data) constructors (including tuple constructors); and *N* is a newtype constructor.

Rule (b) matches a general source-language `case` expression, regardless of whether it actually includes guards – if no guards are written, then `True` is substituted for the guards *g<sub>i,j</sub>* in the *match<sub>i</sub>* forms. Subsequent identities manipulate the resulting `case` expression into simpler and simpler forms.

Rule (h) in Figure 3.2 involves the overloaded operator `==`; it is this rule that defines the meaning of pattern matching against overloaded constants.

These identities all preserve the static semantics. Rules (d), (e), (j), (q), and (s) use a lambda rather than a `let`; this indicates that variables bound by `case` are monomorphically typed (Section 4.1.4).

- (g)  $\text{case } v \text{ of } \{ K p_1 \dots p_n \rightarrow e; \_ \rightarrow e' \}$   
 $= \text{case } v \text{ of } \{$   
 $\quad K x_1 \dots x_n \rightarrow \text{case } x_1 \text{ of } \{$   
 $\quad\quad p_1 \rightarrow \dots \text{case } x_n \text{ of } \{ p_n \rightarrow e; \_ \rightarrow e' \} \dots$   
 $\quad\quad \_ \rightarrow e' \}$   
 $\quad \_ \rightarrow e' \}$   
 at least one of  $p_1, \dots, p_n$  is not a variable;  $x_1, \dots, x_n$  are new variables
- (h)  $\text{case } v \text{ of } \{ k \rightarrow e; \_ \rightarrow e' \} = \text{if } (v == k) \text{ then } e \text{ else } e'$   
 where  $k$  is a numeric, character, or string literal.
- (i)  $\text{case } v \text{ of } \{ x \rightarrow e; \_ \rightarrow e' \} = \text{case } v \text{ of } \{ x \rightarrow e \}$
- (j)  $\text{case } v \text{ of } \{ x \rightarrow e \} = (\backslash x \rightarrow e) v$
- (k)  $\text{case } N v \text{ of } \{ N p \rightarrow e; \_ \rightarrow e' \}$   
 $= \text{case } v \text{ of } \{ p \rightarrow e; \_ \rightarrow e' \}$   
 where  $N$  is a newtype constructor
- (l)  $\text{case } \perp \text{ of } \{ N p \rightarrow e; \_ \rightarrow e' \} = \text{case } \perp \text{ of } \{ p \rightarrow e \}$   
 where  $N$  is a newtype constructor
- (m)  $\text{case } v \text{ of } \{ K \{ f_1 = p_1, f_2 = p_2, \dots \} \rightarrow e; \_ \rightarrow e' \}$   
 $= \text{case } e' \text{ of } \{$   
 $\quad y \rightarrow$   
 $\quad \text{case } v \text{ of } \{$   
 $\quad \quad K \{ f_1 = p_1 \} \rightarrow$   
 $\quad \quad \quad \text{case } v \text{ of } \{ K \{ f_2 = p_2, \dots \} \rightarrow e; \_ \rightarrow y \};$   
 $\quad \quad \_ \rightarrow y \}$   
 $\quad \_ \rightarrow y \}$   
 where  $f_1, f_2, \dots$  are fields of constructor  $K$ ;  $y$  is a new variable
- (n)  $\text{case } v \text{ of } \{ K \{ f = p \} \rightarrow e; \_ \rightarrow e' \}$   
 $= \text{case } v \text{ of } \{$   
 $\quad K p_1 \dots p_n \rightarrow e; \_ \rightarrow e' \}$   
 where  $p_i$  is  $p$  if  $f$  labels the  $i$ th component of  $K$ ,  $\_$  otherwise
- (o)  $\text{case } v \text{ of } \{ K \{ \} \rightarrow e; \_ \rightarrow e' \}$   
 $= \text{case } v \text{ of } \{$   
 $\quad K \_ \dots \_ \rightarrow e; \_ \rightarrow e' \}$
- (p)  $\text{case } (K' e_1 \dots e_m) \text{ of } \{ K x_1 \dots x_n \rightarrow e; \_ \rightarrow e' \} = e'$   
 where  $K$  and  $K'$  are distinct data constructors of arity  $n$  and  $m$ , respectively
- (q)  $\text{case } (K e_1 \dots e_n) \text{ of } \{ K x_1 \dots x_n \rightarrow e; \_ \rightarrow e' \}$   
 $= (\backslash x_1 \dots x_n \rightarrow e) e_1 \dots e_n$   
 where  $K$  is a data constructor of arity  $n$
- (r)  $\text{case } \perp \text{ of } \{ K x_1 \dots x_n \rightarrow e; \_ \rightarrow e' \} = \perp$   
 where  $K$  is a data constructor of arity  $n$
- (s)  $\text{case } v \text{ of } \{ x+k \rightarrow e; \_ \rightarrow e' \}$   
 $= \text{if } v \geq k \text{ then } (\backslash x \rightarrow e)(v-k) \text{ else } e'$   
 where  $k$  is a numeric literal

Figure 3.2: Semantics of Case Expressions, Part 2